MODEL DE CREŞTERE ECONOMICĂ RAMSEY

Neculai Andrei

Institutul Naţional de Cercetare – Dezvoltare
în Informatică, ICI, Bucureşti

Rezumat. Prezentăm modelul de creştere economică Ramsey, precum şi reprezentarea şi soluţia acestuia în limbajul GAMS.

Cuvinte cheie: Ramsey, creştere economică, GAMS.

1. Modelul Ramsey

Teoria creşterii economice se bazează pe lucrările de pionierat ale lui Frank Ramsey [1928], care în scurt timp au devenit un instrument standard în modelarea economică aplicată. În domeniul economic, Ramsey a publicat două lucrări: „A Contribution to the Theory of Taxation" şi „A Mathematical Theory of Saving". Acestea au avut un impact deosebit asupra teoriei matematice a economiei, determinând utilizarea modelelor din ce în ce mai sofisticate în acest domeniu.

Pentru a defini un model de creştere economică să introducem: C_t consumul la momentul de timp t şi ρ rata de discaunt. Fie în acelaşi timp $u(C_t)$ o funcţie de utilitate instantanee, care se bucură de următoarele proprietăţi: $u(C_t) \geq 0$, $u'(C_t) \succ 0$ şi $u''(C_t) \prec 0$. Cu acestea, maximizarea utilităţii (preferinţa consumatorului) conduce la următoarea funcţie de bunăstare:

$$maxW = \int_{t=0}^{\infty} u(C_t) e^{-\rho t} \mathrm{d}t.$$



Frank Ramsey (1903-1930)

În continuare, să introducem $f(K_t)$ funcţia de producţie dependentă de capitalul K_t, cu următoarele proprietăţi: $f(0) = 0$, $f'(0) = \infty$ şi $f'(\infty) = 0$, (Inada [1963] conditions).

Cu acestea, ecuaţia de acumulare a capitalului se defineşte prin intermediul restricţiei bugetare sub forma:

$$f(K_t) = C_t + \frac{\mathrm{d}K}{\mathrm{d}t},$$

unde termenul $\mathrm{d}K / \mathrm{d}t$ reprezintă investiţia netă. Cu alte cuvinte, ieşirea este consumată sau investită.

Studiul analitic al acestui model este prezentat în [8], [9], [18]. Pentru studiul comportării numerice vom introduce o variantă discretă a modelului de mai sus sub forma [19], [14]:

$$maxW = \sum_{t=0}^{\infty} \left(\frac{1}{1+\rho}\right)^t u(C_t)$$

referitor la:

$$Y_t = f(K_t)$$
$$Y_t = C_t + I_t$$
$$K_{t+1} = (1-\delta)K_t + I_t,$$

unde δ este rata de depreciere a capitalului.

De obicei, funcţia de utilitate $u(.)$ şi funcţia de producţie $f(.)$ se definesc sub forma:

$$u(C_t) = \log(C_t),$$
$$f(K_t) = aK_t^b L_t^{1-b},$$

care este o funcţie de tip Cobb-Douglas, unde L_t este factorul de producţie determinat sub forma:

$$L_{t+1} = (1+g)L_t,$$

unde g este rata de creştere a producţiei.

Pentru a utiliza acest model, este necesar să tratăm orizontul de timp infinit din funcţia obiectiv. O soluţie foarte simplă şi la îndemână este da a ignora termenii seriei de la un anumit moment de timp $t = T$. Totuşi, o soluţie mai rafinată consideră $C_t = C_T$ pentru $t > T$, adică de la un moment T încolo consumul este constant.

În acest moment, să introducem o mică digresiune. După cum ştim, pentru orice $0 < a < 1$, putem scrie:

$$\sum_{t=0}^{\infty} a^t = \lim_{T\to\infty} \sum_{t=0}^{T-1} a^t = \lim_{T\to\infty} \frac{1-a^T}{1-a} = \frac{1}{1-a}.$$

Dar,

$$\sum_{t=0}^{T-1} a^t = \frac{1-a^T}{1-a}$$

şi deci,

$$\sum_{t=T}^{\infty} a^t = \sum_{t=0}^{\infty} a^t - \sum_{t=0}^{T-1} a^t = \frac{1}{1-a} - \frac{1-a^T}{1-a} = \frac{a^T}{1-a}.$$

Cu acestea, funcţia obiectiv devine;

$$\begin{aligned} maxW &= \sum_{t=0}^{T-1}\left(\frac{1}{1+\rho}\right)^t \log(C_t) + \sum_{t=T}^{\infty}\left(\frac{1}{1+\rho}\right)^t \log(C_T) \\ &= \sum_{t=0}^{T-1}\left(\frac{1}{1+\rho}\right)^t \log(C_t) + \frac{1}{\rho(1+\rho)^{T-1}} \log(C_T) \\ &= \sum_{t=0}^{T} \beta_t \log(C_t), \end{aligned}$$

unde:

$$\beta_t = \begin{cases} (1+\rho)^{-t} &, \quad t < T, \\ \rho^{-1}(1+\rho)^{1-T}, & \quad t = T. \end{cases}$$

Observăm că, astfel formulată, problema asociază o pondere valorii terminale $u(C_T)$. În experimentele de mai jos, vom vedea că această pondere este importantă. Mai mult, acest mod de abordare a problemelor de optimizare cu orizont de timp infinit se poate aplica şi în alte contexte.

Trunchierea funcţiei obiectiv impune introducerea unei restricţii la momentul final T, referitoare la cerinţa unei investiţii minime, cu care să se restarteze mecanismul economic:

$$I_T \geq (g+\delta)K_T,$$

unde, după cum am văzut, g este rata de creştere a producţiei şi δ este rata de depreciere a capitalului.

Cu acestea un model de creştere economică de tip Ramsey se prezintă sub forma:

$$maxW = \sum_{t=0}^{T} \beta_t \log(C_t),$$

referitor la:

$$\begin{aligned} &Y_t = aK_t^b\, L_t^{1-b}, && t = 0,1,\ldots,T, \\ &Y_t = C_t + I_t, && t = 0,1,\ldots,T, \\ &K_{t+1} = (1-\delta)K_t + I_t, && t = 0,1,\ldots,T-1, \\ &I_T \geq (g+\delta)K_T, \end{aligned}$$

unde ρ (factorul de discaunt), g (rata de creştere a producţiei), δ (factorul de depreciere a capitalului), K_0 (capitalul iniţial), I_0 (investiţia iniţială), C_0 (consumul iniţial), L_0 (factorul de producţie iniţial), b şi a (coeficienţi Cobb-Douglas) sunt parametri cunoscuţi.

Observăm imediat că se poate determina coeficientul Cobb-Douglas a sub forma următoare. Avem: $Y_0 = C_0 + I_0$ şi $Y_0 = f(K_0, L_0) = a K_0^b L_0^{1-b}$ deci:

$$a = \frac{C_0 + I_0}{K_0^b L_0^{1-b}}.$$

2. Expresia modelului în GAMS

```
$ontext
 Model de creştere economică (echilibru dinamic)Ramsey
 Bibliografie:
  Frank P. Ramsey, (1928) A Mathematical Theory of Saving,
  Economics Journal, December 1928.
  Erwin Kalvelagen, (2003) An elementary Ramsey growth model.
  http://www.gams.com/~erwin/micro/growth.gms
$offtext
*
* Date asociate modelului:
*
set  t    'perioadele de timp'        / t1*t50 /

SCALARS
    rho   'factorul de discaunt'                     / 0.04 /
    g     'rata de creştere a producţiei'            / 0.03 /
    delta 'factorul de depreciere al capitalului'    / 0.02 /
    K0    'capitalul iniţial'                        / 3.00 /
    I0    'investiţia iniţială'                      / 0.07 /
    C0    'consumul iniţial'                         / 0.95 /
    L0    'factorul iniţial de producţie '            / 1.00 /
    b     'coeficient Cobb Douglas'                  / 0.25 /
    a     'coeficient Cobb Douglas'
;
*
* Mulţimi de lucru
*
SETS
   tfirst(t)    'primul interval (t0)'
   tlast(t)     'ultimul interval (T)'
   tnotlast(t) 'toate intervalele de timp fără ultimul'
;
tfirst(t)$(ord(t)=1) = yes;
tlast(t)$(ord(t)=card(t)) = yes;
tnotlast(t)= not tlast(t);
*
PARAMETERS
   L(t)       'factorul de producţie'
   beta(t)    'ponderea utilităţilor în funcţia obiectiv'
   tval(t)    'valoarea numerica a lui t'
;
tval(t) = ord(t)-1;
*
* Calculul ponderilor utilităţilor.
* Ponderea asociata ultimului interval de timp (ponderea
* terminală)
* compensează folosirea utilităţilor dincolo de momentul de
* timp T.
*
beta(tnotlast(t)) = power(1+rho,-tval(t));
```

```
beta(tlast(t)) = (1/rho)*power(1+rho,1-tval(t));
display beta;
*
* Calculul factorului de producție utilizând un proces
* de creștere exponențială.
*
L(t) = power(1+g,tval(t)) * L0;
*
* Calculul coeficientului Cobb-Douglas a.
*
a = (C0 + I0) / (K0**b * L0**(1-b));
*
VARIABLES
      C(t)        'consumul'
      Y(t)        'producția'
      K(t)        'capitalul'
      I(t)        'investiția'
      W           'utilitatea totala'
;
*
EQUATION
  utilitate        'funcția obiectiv, utilitatea'
  producție(t)     'funcția de producție Cobb-Douglas'
  alocare(t)       'ecuația producției'
  acumulare(t)     'ecuația acumulării capitalului'
  final(t)         'investiția în ultimul interval'
;
utilitate..                W =e= sum(t, beta(t)*log(C(t)));
producție(t)..             Y(t) =e= a * (K(t)**b) * (L(t)**(1-b));
alocare(t)..                  Y(t) =e= C(t) + I(t);
acumulare(tnotlast(t))..   K(t+1) =e= (1-delta)*K(t) + I(t);
final(tlast)..             I(tlast) =g= (g+delta)*K(tlast);
*
* Margini asupra unor variabile pentru a putea evalua funcțiile.
*
K.lo(t) = 0.001;
C.lo(t) = 0.001;
*
* Condițiile inițiale.
*
K.fx(tfirst) = K0;
I.fx(tfirst) = I0;
C.fx(tfirst) = C0;

model ramsey /all/;
solve ramsey maximizing W using nlp;
*
* Afișarea soluției
*
file res1 /growth.txt/
put res1;
put /"timp    C(t)  Y(t)  K(t)  I(t) "/;
loop(t, put t.tl:6, C.l(t):6, Y.l(t):6, K.l(t):6, I.l(t):6 /;);

*End of model
```

Soluția modelului, corespunzătoare datelor precizate, este prezentată în figura 1.



Figura 1. Soluția modelului Ramsey

Expresia GAMS a modelului conține instrucțiunea de afișare a ponderilor β_t din funcția obiectiv (display beta;). Valorile acestor ponderi sunt următoarele:

```
----      51 PARAMETER beta   ponderea utilităților în funcția obiectiv

t1  1.000,     t2  0.962,     t3  0.925,     t4  0.889,     t5  0.855,     t6  0.822
t7  0.790,     t8  0.760,     t9  0.731,     t10 0.703,     t11 0.676,     t12 0.650
t13 0.625,     t14 0.601,     t15 0.577,     t16 0.555,     t17 0.534,     t18 0.513
t19 0.494,     t20 0.475,     t21 0.456,     t22 0.439,     t23 0.422,     t24 0.406
t25 0.390,     t26 0.375,     t27 0.361,     t28 0.347,     t29 0.333,     t30 0.321
t31 0.308,     t32 0.296,     t33 0.285,     t34 0.274,     t35 0.264,     t36 0.253
t37 0.244,     t38 0.234,     t39 0.225,     t40 0.217,     t41 0.208,     t42 0.200
t43 0.193,     t44 0.185,     t45 0.178,     t46 0.171,     t47 0.165,     t48 0.158
t49 0.152,     t50 3.805
```

Observăm că ponderea asociată momentului final β_{50} are o valoare substanțială. Aceasta arată încă o dată importanța tehnicii de transformare a seriei infinite din funcția obiectiv într-o sumă finită.

O variantă de model Ramsey cu 30 de variabile și 21 de restricții este prezentată în [7, pag. 398], unde se prezintă și soluția acestuia obținută cu pachetul SPENBAR [1], [2],[3],[4],[5], [6].

Modele dinamice de tip Ramsey sunt foarte des utilizate în contextul modelării economice. Putem menționa modelul *DICE - Dynamic Integrated Climate-Economy* dezvoltat de Nordhaus [17]; modelul *TAVF - Transitional Alternative Fuels and Vehicles* prezentat de Leiby și Rubin [15] și *MEEN - Modeling the Environment-Economy Nexus* al lui Harrison [12]. O clasă specială de modele, bazată pe teoria modelelor dinamice Ramsey, este dată de modelele *MARKAL-MACRO* [16].

Bibliografie

1. **ANDREI, N.**: Computational Experience with a Modified Penalty-barrier Method for Large-scale Nonlinear Constrained Optimization. Working Paper No. AMOL-96-1, Research Institute for Informatics, ICI, Bucharest, February 6, 1996a.
2. **ANDREI, N.**: Computational Experience with a Modified Penalty-barrier Method for Large-scale Nonlinear Constrained Optimization Technical Paper No. AMOL-96-2, Research Institute for Informatics, ICI, Bucharest, February 12, 1996b.

3. **ANDREI, N.**: Computational Experience with "SPENBAR" a Sparse Variant of a Modified Penalty-barrier Method for Large-scale Nonlinear, Equality and Inequality, Constrained Optimization. Working Paper No. AMOL-96-3, Research Institute for Informatics, ICI, Bucharest, March 10, 1996c.

4. **ANDREI, N.**: Computational Experience with "SPENBAR" a Sparse Variant of a Modified Penalty-barrier Method for Large-scale Nonlinear, Equality and Inequality, Constrained Optimization. Technical Paper No. AMOL-96-4, Research Institute for Informatics, ICI, Bucharest, March 11, 1996d.

5. **ANDREI, N.**: Numerical Examples with "SPENBAR" for Large-scale Nonlinear, Equality and Inequality, Constrained Optimization with Zero Columns in Jacobian Matrices. Technical Paper No. AMOL-96-5, Research Institute for Informatics, ICI, Bucharest, March 29, 1996e.

6. **ANDREI, N.**: Programarea Matematică Avansată. Teorie, Metode Computaționale, Aplicații. Editura Tehnică - Bucureşti, 1999.

7. **ANDREI, N.**: Modele, Probleme de Test şi Aplicații de Programare Matematică. Editura Tehnică, Bucureşti, 2003.

8. **BARRO, R., SALA-I-MARTIN, X.**: Economic Growth, MIT Press, 1998.

9. **BARRO, R.**: Notes on Growth Accounting. Harward University Technical Report, December 17, 1998.

10. **BARRO, R.**: Ramsey Meets Laibson in the Neoclassical Growth Model. Harward University Technical Report, September, 1999.

11. **McKINNEY, D., A. SAVITSKY**: Basic Optimization Models for Water and Energy Management, The University of Texas at Austin Thechnical Report, 2003.

12. **HARRISON, G. W.**: MEEN, Modeling the Environment - economy Nexus. http://theweb.badm.sc.edu/glenn/meen_mod.htm, 2001.

13. **KEN-ICHI INADA:** On a Two-sector Model of Economic Growth: Comments and a Generalization. Review of Economic Studies 30, 1963, pp. 119-127.

14. **KALVELAGEN, E.**: An Elementary Ramsey Growth Model. GAMS Technical Report, March 12, 2003.

15. **LEIBY, P., J. RUBIN:** Technical Documentation of the Transitional Alternative Fuels and Vehicles (TAFV) Model. Technical Report, Oak Ridge National Laboratory, July 1997.

16. **MANNE, A., C. WENE:** MARKAL-MACRO: a Linked Model for Energy-economy Analysis. Technical Report BNL-47161, Brookhaven National Laboratory, 1992.

17. **NORDHAUS, W.D.**: An Optimal Transition Path for Controlling Greenhouse Gasses. Science 258, 1992, pp. 1315-1319.

18. **PEDERSEN, T.M.:** The Ramsey Model of Optimal Economic Growth. Technical Report, Institute of Economics, University of Copenhagen, December 1999.

19. **RAMSEY, F.P.**: A Mathematical Theory of Saving. Economic Journal, Vol. 38, No. 152, 1928, pp. 543-559.